

REDEFMOVC

par André Chéramy

Présentation :



REDEFMOVC.COM permet de redéfinir tous les caractères du **jeu n°0 (caractères standards ou «normaux»)** et du **jeu n°1 (caractères alternés ou «semi graphiques»)**, ainsi que de déplacer des caractères à l'intérieur d'un jeu ou d'un jeu à l'autre.

Pour le jeu n°0, ces caractères vont du **code ASCII 32 (espace) au code ASCII 127 (carré de la couleur de l'encre)**. Attention : PRINT CHR\$(127) n'affiche pas le caractère de code 127 mais mime un DEL. Toutefois, PLOT x,y,127 ou POKE#BBAA,127 ou CHAR 127,0,1 marchent normalement, c'est à dire affichent un ■.

Les caractères du jeu n°1 n'existent que du **code ASCII 32 au code ASCII 95 inclus**. Il est cependant possible de créer avec REDEFMOVC, des caractères allant de 96 à 111 inclus. Ces 16 caractères «clandestins» sont parfaitement utilisables, toutefois ils ne sont pas régénérables par la routine ROM #F8D0. NB : **le caractère de code 96 est lui même utilisé par REDEFMOVC.**

Package :

Il comprend les 4 fichiers : REDEFMOVC.COM Programme proprement dit. REDEFMOVC.HLP Page écran d'informations/aide. REDEFMOVC.DEM Démonstration réalisée à l'aide d'un jeu de caractères redéfinis. REDEFMOVC.CHS exemple de jeu redéfini.



Mode d'emploi :


- 1) A la question «**Jeu (0/1/ESC)?**» répondre 0 pour redéfinir un caractère normal, 1 pour redéfinir un caractère semi graphique ou en appuyant sur la touche ESCAPE pour obtenir l'affichage de tous les caractères, tels qu'ils sont actuellement définis.
- 2) A la question «**Caractère ?**» répondre en appuyant soit sur la touche correspondant au caractère que vous voulez redéfinir, soit sur la touche **ESCAPE** si le caractère en question n'est pas accessible par le clavier. Dans ce dernier cas, indiquez alors le **code ASCII** du caractère.



- 4) Promenez vous dans la grille 6X8 à l'aide des **4 flèches**, en modifiant le caractère à votre grès: **C** (comme Clair) pour effacer un point ou **N** (comme Noir) pour rajouter un point. **ENTER** pour terminer la redéfinition de ce caractère.

- 5) Si vous avez **un seul caractère** à redéfinir, récupérez le sous programme indiqué avec **CTRL/A**, éliminez REDEFMOVC avec un **DELETE 1000 2500** et sauvez le reste avec **SAVE«DATA»**. A joindre ensuite à votre programme avec un **LOAD«DATA»,J**.

6) Si vous redéfinissez **beaucoup de caractères**, les nouvelles définitions s'accumulent en RAM et peuvent être sauvegardées d'un seul coup en repassant à l'aide de **CTRL/A** sur les lignes «SAVE...»



affichées par REDEFMOVC. Les deux jeux peuvent bien sûr être sauvegardés d'un seul coup en faisant: **SAVE«NOM.CHS»,A#B400,E#BB00**. NB: dans tous les cas où vous avez redéfini des caractères du **jeu 1 au delà du code 95**, il faudra modifier en conséquence l'adresse de fin, qui pourra aller **jusqu'à #BB7F** pour le code ASCII 111 (maximum).

7) Il est possible de restaurer tous les caractères du jeu 0 en faisant **ACCENT OFF** ou ceux du jeu 0 plus ceux du jeu 1 (dont le code est compris entre 32 et 95 inclus) en faisant: **CALL#F8D0** suivi éventuellement de: **ACCENT SET** (qui ne régénère que les 6 caractères «accentués»). Pour l'Oric-1 utilisez **CALL#F89B** pour régénérer les deux jeux de caractères.

Résumé des Jeux, Codes et zone de stockage en RAM:

Jeu standard (96 caractères normaux):

Ces caractères conformes au standard ASCII sauf ceux de code: 94(↑), 95(£), 96(©), 126(#) et 127(■). Attention, un **PRINT CHR\$(127)** n'affiche pas le caractère 127 mais produit un DEL. Par contre **CHAR 127,0,1** ou **PLOTx,y,127** ou **POKE5 5 #BBAA,127** affichent bien le caractère ■ présent en RAM. Après toute redéfinition, **ACCENT OFF** régénère les caractères originaux.

Jeu	ACCENT	Code ASCII	Stockage RAM
0	OFF	32 à 127	#B500 à B7FF

Jeu standard (idem ci dessus, avec 6 caractères «accentués») : Contrairement à la norme ISO qui comporte une douzaine de caractères de substitution, ce jeu ne diffère du précédent que par les 6 caractères de code: 64 (à), 92 (ç), 123 (é), 124 (ù), 125 (è) et 126 (ê). Il peut être régénéré par **CALL#F8D0** suivit de **ACCENT SET**.

Jeu	ACCENT	Code ASCII	Stockage RAM
0	SET	32 à 127	#B500 à B7FF

Jeu alterné (64 caractères semi graphiques) :

Ce jeu ne comporte que 64 caractères. Ces caractères, qui sont spécifiques Oric, peuvent être régénérés, en même temps que le jeu 0, par **CALL#F8D0**.

Jeu	ACCENT	Code ASCII	Stockage RAM
1	néant	32 à 95	#B900 à BAFF

Jeu alterné (extension non officielle de 16 caractères) :

Jeu	ACCENT	Code ASCII	Stockage RAM
1	néant	96 à 111	#BB00 à BB7F

Le jeu alterné peut être étendu de 16 caractères «clandestins» utilisables normalement, mais non régénérés par **CALL#F8D0**. Ces caractères sont définis dans la zone mémoire inutilisée juste avant l'écran **TEXT**.

NB1 : Tout caractère du Jeu **J** et de Code ASCII **C** est défini par 8 octets situés en RAM de l'adresse : **#B400 + #400*J + #8*C** à l'adresse : **#B400 + #400*J + #8*C + #7** ou encore en décimal : de l'adresse: **46080 + 1024*J + 8*C** à l'adresse: **46080 + 1024*J + 8*C + 7**.

NB2 : Le jeu n°0 s'obtient avec **ESC H** ou **PRINT CHR\$(136)**

Le jeu n°1 s'obtient avec **ESC I** ou **PRINT CHR\$(137)**

Exemple de caractères redéfinis

Le jeu des caractères semi graphiques redéfinis, utilisés pour la démonstration **REDEFMOVC.DEM** est disponible sous le nom **REDEFMOVC.CHS**. La correspondance entre code ASCII, caractère du jeu 0 (normal) et caractère du jeu 1 (éléments de cadre) est la suivante :

Lignes simples

Code ASCII	Jeu 0	Jeu 1
33	!	+
34	")
35	#	,
36	\$	*
37	%	-
38	&	.
39	'	3
40	(1
41)	2
42	*	0
43	+	/

Lignes doubles

Code ASCII	Jeu 0	Jeu 1
44	,	6
45	-	4
46	.	7
47	/	5
48	0	8
49	1	9
50	2	>
51	3	<
52	4	=
53	5	;
54	6	:

V. simples & H. doubles

Code ASCII	Jeu 0	Jeu 1
55	7	?
56	8	4
57	9	@
58	:	*
59	;	A
60	<	B
61	=	P
62	>	I
63	?	N
64	@	L
65	A	G

V. doubles & H. simples

Code ASCII	Jeu 0	Jeu 1
66	B	C
67	C)
68	D	D
69	E	5
70	F	E
71	G	F
72	H	0
73	I	M
74	J	J
75	K	H
76	L	K

Le listing

```

1000 ' ***** REDEFMOVC *****
1010 ' adapt{ par A. Ch{ramy 1990
1020 DIM F(8):GOSUB 2010
1030 CLS:PAPER1:INK3:PRINT:PRINT:PRINT
1040 PRINT:PRINT:PRINT"   Red{finition de caract}res (R) "
1050 PRINT:PRINT"   D{placement de caract}res (D) "
1060 PRINT:PRINT"   Instructions (I) "
1070 PRINT:PRINT"   Fin (F) "
1080 PRINT:PRINT:PRINT"           Votre choix ?";:GET R$:PRINT R$
1090 IF R$="R"THEN 1150
1100 IF R$="D"THEN 1700
1110 IF R$="I"THEN 1140
1120 IF R$="F"THEN 1880
1130 GOTO 1030

```

```

1140 !REDEFMOVC.HLP:GET R$:GOTO 1030
1150 PAPER3:INK4:HIRES:PAPER3:INK4
1160 FOR X=100 TO 136 STEP 6:CURSET X,40,1:DRAW 0,48,1:NEXT X
1170 FOR Y=40 TO 88 STEP 6:CURSET 100,Y,1:DRAW 36,0,1:NEXT Y
1180 CLS:PRINT"Jeu (0/1/ESC)?"::GOSUB 2060
1190 PRINT"Caract}re ?":GOSUB 2110
1200 PRINT"Carat}re"CHR$(136+J)CHR$(CT)CHR$(136)"de code ASCII"CT"jeu"J
1210 PRINT"C pour point Clair, N pour point Noir"
1220 POKE#26A,(PEEK(#26A)AND#FE)
1230 PRINT"      Utilisez les 4 fl}ches";
1240 CURSET 40,40,0:CHAR CT,J,1
1250 FOR Y=40 TO 47:FOR X=40 TO 45
1260 IF POINT (X,Y)=-1 THEN CURSET X*6-139,Y*6-200,3:CHAR 96,1,1
1270 NEXT X:NEXT Y
1280 DEF FNC(C)=C*6+95
1290 DEF FNL(L)=L*6+34
1300 L=1:C=1
1310 REPEAT
1320 A=POINT(C+39,L+39):CURSET FNC(C),FNL(L),3
1330 CHAR 96,1,-NOT A:WAIT 5
1340 GOSUB 1580
1350 CHAR 96,1,-A:WAIT 5
1360 UNTIL T$=CHR$(13)
1370 FOR L=1 TO 8:F(L)=0
1380 FOR C=1 TO 6:F(L)=2*F(L)-POINT(C+39,L+39)
1390 NEXT C:NEXT L
1400 TEXT:PRINT
1410 PRINT"Carat}re"CHR$(136+J)CHR$(CT)CHR$(136)"de code
ASCII"CT"jeu"J:PRINT
1420 PRINT"Voici le sous-programme n{c{ssaire"
1430 PRINT"pour obtenir ce caract}re:":PRINT
1440 PRINT"10000 ";:IF J=0 THEN PRINT"J=0"; ELSE PRINT"J=1";
1450 CT$=STR$(CT):CT$=RIGHT$(CT$,LEN(CT$)-1)
1460 PRINT":CT="CT$:AD=#B400+#400*J+8*CT"
1470 PRINT"10010 RESTORE 10030
1480 PRINT"10020 FORADTOAD+7:READV:POKEAD,V:NEXT
1490 PRINT"10030 DATA ";
1500 FOR L=1 TO 8:D$=STR$(F(L)):PRINTRIGHT$(D$,LEN(D$)-1)", ";
1510 POKE 46080+1024*J+CT*8+L-1,F(L)
1520 NEXT L:PRINTCHR$(8) " "
1530 PRINT" ("CHR$(136+J)CHR$(CT)CHR$(136)"sera pok{ entre ";
1540 PRINT HEX$(#B400+#400*J+CT*8) " et "HEX$(#B400+#400*J+CT*8+7) " ) "
1550 POKE#26A,(PEEK(#26A)OR#01)
1560 PRINT:PRINT"Voulez-vous encore red{finir (O/N)?"::GET R$
1570 PRINT R$:IF R$="O" THEN 1150 ELSE 1890
1580 T$=KEY$
1590 IF T$<>" THEN CODE=ASC(T$) ELSE RETURN
1600 IF T$="C" THEN CURSET C+39,L+39,0:CURSET FNC(C),FNL(L),3
1610 IF T$="N" THEN CURSET C+39,L+39,1:CURSET FNC(C),FNL(L),3
1620 IF CODE<8 OR CODE>11 THEN RETURN
1630 C=C+(CODE=8)-(CODE=9)
1640 L=L-(CODE=10)+(CODE=11)
1650 IF C<1 THEN C=1
1660 IF C>6 THEN C=6
1670 IF L<1 THEN L=1
1680 IF L>8 THEN L=8
1690 RETURN
1700 CLS:PAPER4:INK3:PRINT:PRINT"      Permet de d{placer un caract}re"
1710 PRINT"      @ l'interieur d'un m"CHR$(126)"me jeu"
1720 PRINT"      ou d'un jeu @ l'autre.":PRINT:PRINT:PRINT:PRINT
1730 PRINT"      Jeu de d{part (0/1/ESC)?"::GOSUB 2060:JD=J
1740 PRINT"      Caract}re de d{part}?"::GOSUB 2110:CD=CT
1750 PRINT
1760 PRINT"      Jeu d'arriv{e (0/1/ESC)?"::GOSUB 2060:JA=J
1770 PRINT"      Caract}re d'arriv{e?"::GOSUB 2110:CA=CT
1780 PRINT@2,17;"Le carat}re"CHR$(136+JA)CHR$(CA)CHR$(136);
1790 PRINT"de code ASCII"CA" jeu"JA

```

```

1800 PRINT"sera {cras{! OK (O/N)?";:GET R$:IF R$<>"O" THEN 1700
1810 FOR L=1 TO 8
1820 V=PEEK(46080+1024*JD+CD*8+L-1)
1830 POKE46080+1024*JA+CA*8+L-1,V
1840 NEXT
1850 PRINT:PRINT"Voil@, c'est fait!"
1860 PRINT:PRINT:PRINT"Voulez-vous encore d{placer (O/N)?";
1870 GET R$:IF R$="O" THEN 1700
1880 CLS:PAPER1:INK3:PRINT:PRINT:PRINT
1890 PRINT@2,12;"Rappel: " :PRINT
1900 PRINT"SAVE"CHR$(34)"JEU0.CHS"CHR$(34)",A#B500,E#B800"
1910 PRINT"(Sauve le jeu standard en entier)"
1920 PRINT:PRINT"SAVE"CHR$(34)"JEU1.CHS"CHR$(34)",A#B900,E#BB00"
1930 PRINT"(Sauve le jeu semi-graph. en entier)"
1940 PRINT:PRINT"CALL#F8D0:ACCENT SET: "
1950 PRINT"(r{g{n}re tous les caract}res)"
1960 PRINT:PRINT"CTRL/A pour utiliser ces informations"
1970 POKE#26A,(PEEK(#26A)OR#01)
1980 PRINT:PRINT"Retour au 1er menu (O/N)?";
1990 GET R$:PRINT R$:IF R$="O" THEN 1030 ELSE 2430
2000 ' Red{finition code 96 du jeu 1
2010 !RESTORE 2030
2020 FOR AD=#BB00 TO #BB07:READV:POKEAD,V:NEXT
2030 DATA 0,62,62,62,62,62,0,0
2040 RETURN
2050 ' S/P Saisie Jeu et Caract}re
2060 GET R$
2070 IF ASC(R$)=27 THEN GOSUB 2210
2080 IF R$<"O" OR R$>"1" THEN PING:GOTO2060
2090 PRINTR$:J=VAL(R$)
2100 RETURN
2110 PRINT"(de ESP @ ";
2120 IF J=0 THEN PRINTCHR$(126); ELSE PRINT"o";
2130 PRINT" ou ESC pour le code)";:GETR$
2140 IF J=1 AND R$>"o" THEN PRINT:PRINT"Code 111 maxi pour le jeu 1":GOTO
2160
2150 IF ASC(R$)<>27 THEN PRINTR$:CT=ASC(R$):GOTO 2190
2160 PRINT:PRINT"Jeu"J" Code ASCII (de 32 @ ";
2170 IF J=0 THEN PRINT"127)"; ELSE PRINT"95)";
2180 INPUT CT
2190 IF CT<32 OR (J=0 AND CT>127) OR (J=1 AND CT>111) THEN PING:GOTO 2060
2200 RETURN
2210 ' Affichage Jeu 0 et 1
2220 IF PEEK(#21F)<>0 THEN 2320
2230 T1$=" !"+CHR$(34)+"#$%&'()*+,-./0123456789:;<=>?"
2240 T2$="@ABCDEFGHIJKLMNORSTUVWXYZ[\]^_"
2250 T3$=CHR$(96)+"abcdefghijklmnopqrstuvwxy{|}"+CHR$(126)+CHR$(127)
2260 PLOT3,1,8:PLOT4,1,T1$
2270 PLOT3,2,8:PLOT4,2,T2$
2280 PLOT3,3,8:PLOT4,3,T3$
2290 PLOT3,5,9:PLOT4,5,T1$:PLOT36,5,8
2300 PLOT3,6,9:PLOT4,6,T2$:PLOT36,6,8
2310 RETURN
2320 FOR J=0 TO 1
2330 FOR L=1 TO 3
2340 CURSET25,120+10*L+30*J,0
2350 FOR AS=32*L TO 32*L+31
2360 CHARAS,J,1
2370 CURMOV6,0,0
2380 IF J=1 AND AS=95 THEN 2420
2390 NEXT AS
2400 NEXT L
2410 NEXT J
2420 RETURN
2430 PRINT@2,25;" "
2440 PRINTCHR$(11)CHR$(11)CHR$(11):END

```